# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=flashgbx
pkgver=3.34
pkgrel=1
pkgdesc='Reads and writes Game Boy and Game Boy Advance cartridge data'
arch=(any)
url=https://github.com/lesserkuma/FlashGBX
license=(GPL3)
depends=(
  pyside6
  python-dateutil
  python-pillow
  python-pyserial
  python-requests
  python-setuptools
)
makedepends=(git)
_tag=c12b1de4dd66a8ec85bab1e0e70bf8661b4255fd
source=(git+https://github.com/lesserkuma/FlashGBX.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd FlashGBX
  git describe --tags
}

build() {
  cd FlashGBX
  python setup.py build
}

package() {
  cd FlashGBX
  python setup.py install --skip-build --root="${pkgdir}" --optimize=1 --skip-build
}

# vim: ts=2 sw=2 et:
